Transition from log to power law in the DOS for the M point singularities in the surface bands of $Sr_2 RuO_4$

Anirudh Chandrasekaran

Loughborough University

17 June 2024

Loading the Library

We first load the library by executing the following:

In[ ]:=
```
AppendTo[$Path, StringDelete[NotebookDirectory[], "/Sr2RuO4"] <> "Package"];
<< BandUtilities`
```

Interpolating between the Wannier90 files

Ingredients needed for symmetrization

In[222]:=
```
ρ = {{0, -1, 0}, {1, 0, 0}, {0, 0, 1}};
(*The matrix part of the C4 rotation about one of the Ru atoms.*)
Uρ = {{0, -1, 0}, {1, 0, 0}, {0, 0, -1}};
(*Representation of C4 rotation in the d-orbital space.*)

rxy = {{0, 1, 0}, {1, 0, 0}, {0, 0, 1}}; (*x↔y reflection.*)
Urxy = {{-1, 0, 0}, {0, 1, 0}, {0, 0, -1}};
(*Representation of the x↔y reflection.*)

c0 = {1/4, 3/4, 0}; (*Location of one of the Ru atoms,
which we shall use as the centre of the C4 rotation.*)

AtomLocations =
  {{3/4, 1/4, 0}, {3/4, 1/4, 0}, {3/4, 1/4, 0}, {1/4, 3/4, 0}, {1/4, 3/4, 0}, {1/4, 3/4, 0}};
(*Location of the basis atoms in the fractional
 coordinates within the unit cell.*)

SymmetryList = {}; (*This is a multidimensional list which we shall
 construct below. Each element is a sublist corresponding to a particular
 symmetry. First element of the sublist is real space symmetry's matrix part,
second element is the translation part and the third element is
 its representation in the Hilbert space. Our point group is D4,
or D8 in some pepole's notation.*)
Do[
  AppendTo[SymmetryList,
   {MatrixPower[ρ, i], (c0 - MatrixPower[ρ, i].c0) // Simplify,
    ArrayFlatten[IdentityMatrix[2]⊗MatrixPower[Uρ, i]]}];

  AppendTo[SymmetryList,
   {MatrixPower[ρ, i].rxy, (c0 - MatrixPower[ρ, i].c0) // Simplify,
    ArrayFlatten[IdentityMatrix[2]⊗MatrixPower[Uρ, i]].
     ArrayFlatten[{{0, 1}, {1, 0}}⊗Urxy]}];
  , {i, 1, 4}];
```

Interpolation with symmetrization

We now interpolate between the Wannier90 data for 13 different rotations of the RuO octahedra: 0 ° through 12° to generate a continuously θ dependent Hamiltonian, whilst ensuring that the interpolated model has the right symmetry (for which we are feeding the list SymmetryList):

```
In[230]:= FileLocn = NotebookDirectory[] <> "unrenormalized2/";

W90Data = LoadInterpolateTBMs[{FileLocn <> "Sr2RuO4_0deg.dat",
    FileLocn <> "Sr2RuO4_1deg.dat", FileLocn <> "Sr2RuO4_2deg.dat",
    FileLocn <> "Sr2RuO4_3deg.dat", FileLocn <> "Sr2RuO4_4deg.dat",
    FileLocn <> "Sr2RuO4_5deg.dat", FileLocn <> "Sr2RuO4_6deg.dat",
    FileLocn <> "Sr2RuO4_7deg.dat", FileLocn <> "Sr2RuO4_8deg.dat",
    FileLocn <> "Sr2RuO4_9deg.dat", FileLocn <> "Sr2RuO4_10deg.dat",
    FileLocn <> "Sr2RuO4_11deg.dat", FileLocn <> "Sr2RuO4_12deg.dat"},
   {0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10, 11, 12}, "Symmetries" → SymmetryList,
   "OrbitalCentres" → AtomLocations, "ParameterSymbol" → "θ"];

Hamltn1[k1_, k2_, θ_] = W90Data⟦1⟧;
```

```
Invalid or no lattice vectors provided! Using unit vectors.

Tight-binding data files loaded.

Space dimension: 3

Number of orbitals: 6

Number of supercells: 49

Number of models/data-files to interpolate: 13

Number of lines in the data files:
 {741, 853, 853, 853, 845, 853, 853, 853, 853, 853, 853, 853, 853}

Interpolating between the Hamiltonians.

Valid symmetrization scheme provided! We will symmetrize the loaded Hamiltonian.

Number of supercells after symmetrization: 79
```

We examine the series expansion at the M point for a particular angle, say 9°

```
In[233]:= HamTempy[k_] := Hamltn1[k⟦1⟧ - k⟦2⟧, k⟦1⟧ + k⟦2⟧, 9]
```

To find the correct band to series expand, we first compute the eigenvalues at the M point

```
In[234]:= Sort[Eigenvalues[HamTempy[{0, π // N}]]]

Out[234]= {-0.535252, -0.535252, -0.090436, -0.090436, 0.289416, 0.289416}
```

Therefore the bands of interest just below the Fermi level are 3 and 4

```
In[235]:= TaylTempy = TaylorExpandBand[HamTempy,
    {0, π // N}, 4, 4, "Messages" → False] /. {p1 → kx, p2 → ky};
NumberForm[Total[TaylTempy⟦1⟧], 2]
NumberForm[Total[TaylTempy⟦2⟧], 2]
```

Out[236]//NumberForm=

$$-0.09 - 0.029\, k_x^2 - 0.4\, k_x^4 + 0.063\, k_y^2 + 0.027\, k_x^2\, k_y^2 - 0.04\, k_y^4$$

Out[237]//NumberForm=

$$-0.09 + 0.063\, k_x^2 - 0.04\, k_x^4 - 0.029\, k_y^2 + 0.027\, k_x^2\, k_y^2 - 0.4\, k_y^4$$

Strategy

As we tune θ, we will find that the series expansion at the M point takes the form $\alpha k_x^2 + \beta k_y^2 + \gamma k_x^4 + \mu k_x^2 k_y^2 + \nu k_y^4$ along with a 90° rotated copy obtained by applying $(k_x, k_y) \to (-k_y, k_x)$.

Since the routine will output the series expansion of the pair of rotated bands together, we need a strategy to isolate a particular band of interest out of the two. Now the series expansion across the range [7°,12°] indicates that $\nu << \gamma$ for one band and $\nu >> \gamma$ for the other one. For example, at $\theta = 9°$, the unrenormalized, SOC-free model has the following series expansion at the M point: $-0.09 - 0.029 k_x^2 + 0.063 k_y^2 - 0.4 k_x^4 + 0.027 k_y^2 k_x^2 - 0.04 k_y^4$ (and its rotated copy). We can see that the coefficient of the k_x^4 term is about ten times the coefficient of the k_y^4 term. This feature persists for all θ in that range. Therefore, to be consistent, let us choose the band with $\nu < \gamma$. The DOS integral is given by

$$g(\epsilon) = \int_{BZ} d^2 k\, \delta\left(\alpha k_x^2 + \beta k_y^2 + \gamma k_x^4 + \mu k_x^2 k_y^2 + \nu k_y^4 - \epsilon\right)$$

To infer the approximate low energy scaling behaviour of the DOS (particularly for a HOVHS), one normally extends the integrals over the entire (k_x, k_y)-plane (picking up a finite error in the process) and rescales the k_x and k_y in an appropriate fashion. When α is “small” (to be qualified below) we will rescale $(k_x, k_y) \to \left(|\epsilon|^{\frac{1}{4}} k_x,\ |\epsilon|^{\frac{1}{2}} k_y\right)$ to obtain

$$g(\epsilon) \sim \int_{BZ} d^2 k\, (|\epsilon|)^{\left(\frac{1}{4}+\frac{1}{2}\right)}\, \delta\left(\alpha |\epsilon|^{\frac{1}{2}} k_x^2 + \beta |\epsilon| k_y^2 + \gamma |\epsilon| k_x^4 + \mu |\epsilon|^{\frac{3}{2}} k_x^2 k_y^2 + \nu |\epsilon|^2 k_y^4 - \epsilon\right)$$

Assume that $\epsilon > 0$ (the arguments can be applied with a slight modification for $\epsilon < 0$). Then, using the scaling property of the delta function we
obtain

$$g(\epsilon) \sim \int d^2 k\, \epsilon^{\left(\frac{1}{4}+\frac{1}{2}-1\right)}\, \delta\left(\alpha \epsilon^{-\frac{1}{2}} k_x^2 + \beta k_y^2 + \gamma k_x^4 + \mu \epsilon^{\frac{1}{2}} k_x^2 k_y^2 + \nu \epsilon k_y^4 - 1\right)$$

Since $[\epsilon] = E$, $[\alpha] = E.l^2$ and $[\gamma] = E.l^4$, we have $\left[\alpha^2 \epsilon^{-1}\right] = E.l^4 = [\gamma]$ where E and l are respectively energy and length units. Thus, $\alpha^2 \epsilon^{-1}$ and γ are comparable since they have the same dimension. In fact, for $\alpha^2 \epsilon^{-1} << \gamma$, we can ignore the k_x^2 term in comparison to the k_x^4 in the DOS integral by further rescaling $k_x \to k_x \big/ \gamma^{\frac{1}{4}}$. (This is what we meant by α being small earlier). For small energies, the terms containing $\epsilon^{\frac{1}{2}}$ and ϵ can also be ignored so that the final DOS integral scales approximately as

$$g(\epsilon) \sim \epsilon^{-\frac{1}{4}} \int d^2 k\, \delta\left(\beta k_y^2 + \gamma k_x^4 + \nu k_y^4 - 1\right),$$

where the integral is approximately ϵ independent. We proceed similarly for $\epsilon < 0$. Thus the crossover from log to power law is controlled by
$|\epsilon| >> \frac{\alpha^2}{\gamma}$.

That is, for energies well above this scale, the quartic term dominates over the quadratic term in the DOS integral giving an approximate power law behaviour, while for energies well below this scale, the quadratic term dominates, giving a logarithmically diverging DOS.

In[244]:=
```
LogPowerTrans = {};
SetSharedVariable[LogPowerTrans];

Do[
 HamCurr[k_] := Hamltn1[k[[1]] - k[[2]], k[[1]] + k[[2]], θcurr];

 Tayl = TaylorExpandBand[HamCurr, {0, π // N}, 4, 4, "Messages" → False];

 If[Abs[D[Tayl[[1, 5]], {p1, 4}]] > Abs[D[Tayl[[1, 5]], {p2, 4}]],
  TaylTrans = Tayl[[1]];
  ,
  TaylTrans = Tayl[[2]];
 ];

 Etrans = 1000
   (Abs[D[TaylTrans[[3]], {p1, 2}]^2 / D[TaylTrans[[5]], {p1, 4}] Factorial[4] / 4]);

 Print["θ = ", θcurr, "°, Tayl = ",
  NumberForm[Chop[Total[TaylTrans /. {p1 → kx, p2 → ky}], 10^-6], {2, 3}],
  "\n\t |ϵtrans - ϵVHS| = ", NumberForm[Etrans, {3, 2}], " meV."];

 AppendTo[LogPowerTrans, {θcurr, Etrans}];

 , {θcurr, 7, 12, 0.5}]
```

$\theta = 7.°$, Tayl $= -0.051 - 0.160\,k_x^2 - 0.330\,k_x^4 + 0.069\,k_y^2 + 0.034\,k_x^2\,k_y^2 - 0.059\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 80.70$ meV.

$\theta = 7.5°$, Tayl $= -0.061 - 0.130\,k_x^2 - 0.350\,k_x^4 + 0.068\,k_y^2 + 0.032\,k_x^2\,k_y^2 - 0.055\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 48.00$ meV.

$\theta = 8.°$, Tayl $= -0.071 - 0.096\,k_x^2 - 0.370\,k_x^4 + 0.066\,k_y^2 + 0.030\,k_x^2\,k_y^2 - 0.050\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 25.00$ meV.

$\theta = 8.5°$, Tayl $= -0.081 - 0.062\,k_x^2 - 0.390\,k_x^4 + 0.064\,k_y^2 + 0.028\,k_x^2\,k_y^2 - 0.045\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 10.10$ meV.

$\theta = 9.°$, Tayl $= -0.090 - 0.029\,k_x^2 - 0.400\,k_x^4 + 0.063\,k_y^2 + 0.027\,k_x^2\,k_y^2 - 0.040\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 2.11$ meV.

$\theta = 9.5°$, Tayl $= -0.099 + 0.003\,k_x^2 - 0.420\,k_x^4 + 0.062\,k_y^2 + 0.025\,k_x^2\,k_y^2 - 0.036\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 0.03$ meV.

$\theta = 10.°$, Tayl $= -0.110 + 0.035\,k_x^2 - 0.430\,k_x^4 + 0.059\,k_y^2 + 0.023\,k_x^2\,k_y^2 - 0.030\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 2.82$ meV.

$\theta = 10.5°$, Tayl $= -0.120 + 0.065\,k_x^2 - 0.440\,k_x^4 + 0.057\,k_y^2 + 0.020\,k_x^2\,k_y^2 - 0.024\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 9.71$ meV.

$\theta = 11.°$, Tayl $= -0.140 + 0.095\,k_x^2 - 0.450\,k_x^4 + 0.056\,k_y^2 + 0.019\,k_x^2\,k_y^2 - 0.020\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 20.00$ meV.

$\theta = 11.5°$, Tayl $= -0.160 + 0.120\,k_x^2 - 0.460\,k_x^4 + 0.057\,k_y^2 + 0.022\,k_x^2\,k_y^2 - 0.021\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 32.60$ meV.

$\theta = 12.°$, Tayl $= -0.150 + 0.150\,k_x^2 - 0.460\,k_x^4 + 0.043\,k_y^2 + 0.014\,k_x^2\,k_y^2 + 0.020\,k_y^4$

$|\epsilon_{trans} - \epsilon_{VHS}| = 49.00$ meV.

In[247]:=
```
TransFunc = Interpolation[LogPowerTrans];
```

In[248]:=
```
Show[ListPlot[LogPowerTrans], Plot[TransFunc[x], {x, 7, 12}]]
```

Out[248]=



In[249]:=
```
PhasePlt = Plot[{0, Quiet[TransFunc[x]], -Quiet[TransFunc[x]]},
   {x, 0, 12}, PlotRange → {{0, 12}, {-20, 20}}, Axes → False,
   Frame → True, Filling → {2 → {{3}, Opacity[0.3, Blue]},
     2 → {Top, Opacity[0.3, Red]}, 3 → {Bottom, Opacity[0.3, Red]}},
   PlotStyle → {Gray, Black, Black}, AspectRatio → 1,
   FrameLabel → {Style["θ(°)", FontSize → 14], Style["E(meV)", FontSize → 14]}];

Show[PhasePlt(*,Plt1d,Plt1c,Plt1a*), Graphics[
  Text[Style["log", FontFamily → "Times", FontSize → 14], {5, 10}]], Graphics[
  Text[Style["power\nlaw\n1/4", FontFamily → "Times", FontSize → 14], {9.5, 12}]]]
```

Out[250]=